



"PascalABC.Net v.3.8.+ НОВЫЙ ПОДХОД"

Щецова Ольга Владимировна,
учитель информатики МОУ «Лицей № 4 г. Дмитрова»

– это **система программирования и язык Pascal** нового поколения для платформы Microsoft .NET


- современные языковые возможности:
модули, классы, перезагрузка операций, лямбда-выражения, срезы ...
- широкий набор функций и методов, в том числе для обработки строк, массивов и последовательностей;
- компактный и понятный код;
- компилятор аналогичен по скорости C#

<http://pascalabc.net/> - официальный сайт PascalABC.Net

<https://stepik.org/course/91781/syllabus> - Александр Осипов курс «PascalABC.NET: современный код»

ПРИМЕРЫ

```
Program z1;  
var a,b,c: integer;  
Begin  
  write('Введите два числа');  
  read(a,b);  
  c := a + b;  
  write(a, '+', b, '=', c);  
end.
```



```
##  
var (a,b) := ReadInteger2('Введите два числа');  
var c := a + b;  
Print($"{a} + {b} = {c}");
```

```
Program z2;  
var n, a, s:integer;  
Begin  
  readln(n);  
  s := 0;  
  for var i := 1 to n do  
  begin  
    read(a);  
    s := s + a;  
  end;  
  write(s);  
end.
```



```
##  
var n := ReadInteger;  
var s := 0;  
loop n do  
  s += RaedInteger;  
s.Print;
```

Часть 1. Что нового?

```
var a: integer;  
var b: int64;  
var c: BigInteger;  
var x: real;
```

```
var a := 5;  
var b := 12345678901234;  
var c := 10bi;  
var x := -5.0;
```

```
//явное преобразование типов  
ИмяТипа (выражение)
```

```
x := integer('x');  
a := BigInteger(2**3);  
b := Real(a);
```

```
//string  
var str1 := 'a' * 100;  
var str2 := 10 * 'xyz';  
var str3 := 'abc' + 100;
```

АРИФМЕТИЧЕСКИЕ ОПЕРАЦИИ

`a**b, a*b, a/b, a div b, a mod b, a+b, a-b`

`// сокращенная запись`

`a += n; a -= n; a *= n; a /= n;`

КОРТЕЖИ

```
var (a, b, c) := (5, 2.5, 'str');  
var (x, y) := ReadReal2;  
var (n, dmin, s) := (ReadInteger, MaxInt, 0);
```

```
##  
// Наибольший общий делитель  
var (a,b) := ReadInteger2;  
while b>0 do  
    (a,b) := (b, a mod b);  
Print('НОД = ', a);
```

```
##  
// Числа Фибоначчи  
var (a,b) := (1,1);  
loop 20 do  
    begin  
        Print(a);  
        (a,b) := (b, a + b);  
    end;
```


ВСТРОЕННЫЕ ПРОЦЕДУРЫ И ФУНКЦИИ

```
swap (x, y) ;  
m := Max (a, b, c...) ;  
m := Min (a, b, c...) ;  
x := Random (a, b) ;  
y := Round (y, n) ;
```

```
// логические  
x.Divs (y)  
x.NotDivs (y)  
x.inRange (a, b)  
x.isEven  
x.isOdd
```

<https://stepik.org/lesson/480466/step/4?unit=471443>
<http://pascalabc.net/downloads/pabcnethelp/index.htm>

<https://stepik.org/lesson/481484/step/9?unit=472582>

БАЗОВЫЕ КОНСТРУКЦИИ. ВВОД/ВЫВОД

```
##  
var (a,b) := ReadInteger2('Введите два числа');  
var c := a + b;  
Print('${a} + {b} = {c}');
```

```
##  
var (x, y) := ReadReal2('Введите координаты точки:');  
var s := sqrt((x**2 + y**2));  
s.Print;
```

```
## //ВВОД массива  
var n := ReadInteger;  
var a := ReadArrInteger(n);
```

УСЛОВНЫЙ ОПЕРАТОР И УСЛОВНАЯ ОПЕРАЦИЯ

```
if <логическое выражение>  
  then <оператор 1>  
  else <оператор 2>;
```

```
<логическое выражение> ? <выражение 1> : <выражение 2>;
```

```
var x := x mod 2 = 0 ? x div 2 : x - 1;
```

УСЛОВНЫЙ ОПЕРАТОР И УСЛОВНАЯ ОПЕРАЦИЯ

$$y = \begin{cases} 2x, x < 0, \\ x, 0 \leq x < 100, \\ 0, x \geq 100 \end{cases}$$

```
##  
L var x := ReadInteger;  
  var y :=  
    if x < 0  
      then 2 * x  
      else if x < 100  
            then x  
            else 0;
```

ЦИКЛЫ

```
loop <сколько раз> do  
  <тело цикла>;
```

```
for var <параметр> := <выражение "от"> to <выражение "до"> do  
  <тело цикла>;
```

```
for var <параметр> := <выражение "от"> downto <выражение "до"> do  
  <тело цикла>;
```

```
while <условие входа> do  
  <тело цикла>;
```

```
repeat  
  <тело цикла>  
until <условие выхода>;
```

ЦИКЛ И ПОСЛЕДОВАТЕЛЬНОСТИ

```
foreach var <переменная> in <последовательность> do  
    <тело цикла>;
```

```
##  
var n := ReadInteger;  
var a := ReadArrInteger(n);  
var (s, m) := (0, MaxInt);  
foreach var k in a do begin  
    if k.isEven  
        then s += k;  
    m := min(m, k);  
end;  
Print(s, m);
```

```
procedure Имя (параметр1; параметр2; ...);  
begin  
    <тело процедуры>  
end;  
Begin  
    <основная программа>  
End.
```

```
##  
procedure Имя (параметр1; параметр2; ...) := <оператор>;  
...  
Имя (параметр1; параметр2; ...); // вызов процедуры
```

ЛЯМБДА-ВЫРАЖЕНИЕ

```
(Параметр1, Параметр2, ...) -> Выражение;
```

```
(a, b) -> sqrt(a * a + b * b);
```


ПРОГРАММЫ И АЛГОРИТМЫ

https://pascalabcnet.github.io/progr_beginners.html

<https://stepik.org/course/91781/syllabus>

Часть 2. Примеры решений заданий ЕГЭ 2021

```
//системы счисления q = 2..36  
nq := ToBase(n, q); //перевод из 10-ой системы счисления  
n.ToBase(q)  
n := Dec(nq, q); //перевод в 10-ую систему счисления  
n := DecBig(nq, q);
```

```
//Комбинаторика  
s.Permutations //количество перестановок символов в строке  
s.Permutations(m)  
s.Combinations(m)
```

<https://stepik.org/lesson/502616/step/1?unit=494341>

```
//теория чисел  
НОД(a,b);  
НОК(a,b);  
Primes(n); //простые числа до n  
Primes(n, m); //простые числа от n до m  
Factorize(n); //разложение на простые множители  
Divisor(n); //все делители числа  
PrimeDivisorCount(n); //количество делителей числа  
Digits(n) //цифры числа
```

<https://stepik.org/lesson/502616/step/1?unit=494341>

ЗАДАНИЕ 2

Логическая функция F задаётся выражением $((x \wedge \neg y) \vee (w \rightarrow z)) \equiv (z \equiv x)$. На рисунке приведён частично заполненный фрагмент таблицы истинности функции F , содержащий неповторяющиеся строки. Определите, какому столбцу таблицы истинности функции F соответствует каждая из переменных x, y, z, w .

?	?	?	?	F
	0	0	1	1
0	1	0	0	1
0			1	1

В ответе напишите буквы x, y, z, w в том порядке, в котором идут соответствующие им столбцы. Буквы в ответе пишите подряд, никаких разделителей между буквами ставить не нужно.

ЗАДАНИЕ 2

```
##  
uses School;  
var f := TrueTable((x, y, z, w) -> (x and not(y)) or (w <= z) = (z = x));  
TrueTablePrint(f, 1, 'xyzw');
```

x	y	z	w	F
0	0	0	0	1
0	1	0	0	1
1	0	1	0	1
1	0	1	1	1
1	1	0	1	1
1	1	1	0	1
1	1	1	1	1

ЗАДАНИЕ 5

На вход алгоритма подаётся натуральное число N . Алгоритм строит по нему новое число R следующим образом.

1. Строится двоичная запись числа N .
2. К этой записи дописываются справа ещё два разряда по следующему правилу:
 - а) складываются все цифры двоичной записи числа N , и остаток от деления суммы на 2 дописывается в конец числа (справа). Например, запись 11100 преобразуется в запись 111001;
 - б) над этой записью производятся те же действия – справа дописывается остаток от деления суммы её цифр на

Полученная таким образом запись (в ней на два разряда больше, чем в записи исходного числа N) является двоичной записью искомого числа R . Укажите такое **наименьшее число N** , для которого **результат** работы данного алгоритма **больше числа 77**. В ответе это число запишите в десятичной системе счисления.

ЗАДАНИЕ 5

```
## uses School;  
for var n := 1 to 10000 do // да хоть до миллиарда!  
begin  
  var s := Bin(n);  
  loop 2 do  
    s += s.CountOf('1') mod 2;  
    if Dec(s, 2) > 77 then  
      begin  
        n.Print;  
        break  
      end;  
    end;  
end;  
  
{  
Результат выполнения программы:  
19  
}
```


ЗАДАНИЕ 12

Дана программа для исполнителя Редактор:

НАЧАЛО

ПОКА нашлось (2222) ИЛИ нашлось (8888)

ЕСЛИ нашлось (2222)

ТО заменить (2222, 88)

ИНАЧЕ заменить (8888, 22)

КОНЕЦ ЕСЛИ

КОНЕЦ ПОКА

КОНЕЦ

Какая строка получится в результате применения приведённой программы к строке, состоящей из 70 идущих подряд цифр 8? В ответе запишите полученную строку.

ЗАДАНИЕ 12

НАЧАЛО

ПОКА нашлось (2222) ИЛИ нашлось (8888)

ЕСЛИ нашлось (2222)

ТО заменить (2222, 88)

ИНАЧЕ заменить (8888, 22)

КОНЕЦ ЕСЛИ

КОНЕЦ ПОКА

КОНЕЦ

```
##  
var s := 70 * '8';  
while True do  
begin  
  var p := Pos('2222', s);  
  if p > 0  
  then s := s.Replace('2222', '88', 1)  
  else  
  begin  
    p := Pos('8888', s);  
    if p > 0  
    then s := s.Replace('8888', '22', 1)  
    else break  
  end  
end;  
s.Print
```

ЗАДАНИЕ 14

Значение выражения $5^{2004} - 5^{1016} - 25^{508} - 5^{400} + 25^{250} - 27$ записали в пятеричной системе счисления. Сколько цифр 4 в такой записи?

```
##  
uses school;  
var n := 5bi**2004 - 5bi**1016 - 25bi**508 - 5bi**400 + 25bi**250 - 27;  
n.ToBase(5).CountOf('4').Print;  
  
//ответ: 1484
```

ЗАДАНИЕ 14

Значение выражения $12^{34} + 7 \cdot 12^{26} - 3 \cdot 12^{16} + 2 \cdot 12^5 + 552$ записали в системе счисления с основанием 12. Сколько различных цифр содержится в этой записи?

```
##
uses school;
var n := (12bi**34 + 7*12bi**26 - 3*12bi**16 + 2*12bi**5 + 552).ToBase(12);
var k := 0;
for var c := '0' to '9' do
    if n.CountOf(c) <> 0
        then k+=1;
for var c := 'A' to 'B' do
    if n.CountOf(c) <> 0
        then k+=1;
k.Print;

//ответ: 8
```

ЗАДАНИЕ 16

Алгоритм вычисления функции $F(n)$ задан следующими соотношениями:

$$F(n) = 1 + 2n \text{ при } n < 5$$

$$F(n) = 2 \cdot (n + 1) \cdot F(n - 2), \text{ если } n \geq 5 \text{ и делится на } 3,$$

$$F(n) = 2 \cdot n + 1 + F(n - 1) + 2 \cdot F(n - 2), \text{ если } n \geq 5 \text{ и не делится на } 3.$$

Чему равно значение функции $F(15)$?

```
##  
function F(n: integer): integer :=  
  if n < 5 then 1 + 2*n  
  else if n.Divs(3) then 2*(n + 1)*F(n - 2)  
    else 2*n + 1 + F(n-1) + 2*F(n - 2);  
  
F(15).Print;  
  
// Ответ: 5158048
```

ЗАДАНИЕ 10

10 С помощью текстового редактора определите, сколько раз, не считая сносок, встречается слово «долг» или «Долг» в тексте романа в стихах А.С. Пушкина «Евгений Онегин». Другие формы слова «долг», такие как «долги», «долгами» и т.д., учитывать не следует. В ответе укажите только число.

Если вдуматься, задание странноватое. Вот это - "не считая сносок". Например Word - достаточно популярный текстовый процессор - не умеет в поиске различать сноски и иные разновидности текста. Предполагается, что нужно отыскать все вхождения искомого контента и визуально в каждом случае решать, относится он к сноскам или нет? В общем, какое-то не до конца проработанное условие.

Может ли PascalABC.NET справиться с этим заданием? Несомненно, поскольку к заданию кроме файла в формате .docx прилагаются файлы и в другом формате, в частности, обычный текст в формате .txt. А такие файлы читать очень просто. Нам всего-то понадобится составить шаблон регулярного выражения для поиска нужного контекста. Шаблон `\b[Дд]олг\b` ищет заключенную между границами слова букву "Д" или "д", за которой следует подстрока "олг".

```
##  
var fn := 'C:\PABCWork.NET\EG32021 Demo\10.txt'; // путь к файлу  
ReadAllText(fn).MatchValues('\b[Дд]олг\b').Count.Print // 1
```

Ответ: 1. Кто бы мог подумать, что в целом произведении слово "долг" встречается единственный раз.

<https://stepik.org/lesson/502617/step/6?unit=494342>



"PascalABC.Net v.3.8.+ НОВЫЙ ПОДХОД"

<http://pascalabc.net/> - официальный сайт PascalABC.Net

https://pascalabcnet.github.io/mydoc_for_EGE.html - для школы и ЕГЭ

<https://stepik.org/course/91781/syllabus> - курс «PascalABC.NET: современный код»,
автор Александр Осипов

<https://kpolyakov.spb.ru/download/sochi2021.ppt> - презентация К.Ю. Полякова для
учителей информатики г. Сочи